15

The Linear Tetrahedron

TABLE OF CONTENTS

§15.1. Introduction

In this Chapter we study the construction of shape functions for three-dimensional solid elements, beginning with the 4-node tetrahedron. We start with this particular element for two reasons: the geometry is the simplest one, and no numerical integration is needed.

§15.2. The Linear Tetrahedron

The linear tetrahedron, shown in Figure 15.1(a), is not used often for stress analysis because of its poor performance.[1] Its main value in structural and solid mechanics is educational: it serves as a vehicle to introduce the basic steps of formulation of 3D solid elements, particularly as regards use of natural coordinate systems and node numbering conventions. It should be noted that 3D visualization is notoriously more difficult than 2D, so we need to proceed somewhat slowly here.



FIGURE 15.1. (a) The linear tetrahedron element: also called the 4-node tetrahedron; (b) Node numbering convention.

§15.2.1. Tetrahedron Geometry

Figure 15.1 shows a typical 4-node tetrahedron. Its geometry is fully defined by giving the location of the four corner nodes with respect to the global RCC system (x, y, z):

$$x_i, \quad y_i, \quad z_i \quad (i = 1,\ 2,\ 3,\ 4). \tag{15.1}$$

The volume measure of the tetrahedron is denoted[2] by $\mathcal{V}$ and is given by the following determinant:

$$\mathcal{V} = \tfrac{1}{6} \det \begin{bmatrix} 1 & 1 & 1 & 1 \\ x_1 & x_2 & x_3 & x_4 \\ y_1 & y_2 & y_3 & y_4 \\ z_1 & z_2 & z_3 & z_4 \end{bmatrix}. \tag{15.2}$$

[1] Derivative of shape functions are constant over the element volume. Strains and stresses recovered in this manner can be highly inaccurate. This makes the element dangerous for stress analysis. On the other hand, when the objective is merely to get values of primary variables, as in thermal analysis and computational gas dynamics, the linear tetrahedron is acceptable.

[2] This symbol (Upsilon) is used to avoid confusion with V, which denotes the volume of a generic body.

This volume is a *signed* quantity. It is positive if the corners are numbered in such a way that the volume is positive. A numbering rule that achieves this goal is as follows:

(I) Pick a corner as initial one. In Figure 15.1(a) this is numbered 1.

(II) Pick a face that will contain the first three corners. The excluded corner will be the last one.

(III) Number these three corners in a *counterclockwise* sense when looking at the face from the excluded corner. See Figure 15.1(b).

In what follows we shall always assume that the numbering has been done in that manner so that $\mathcal{V} > 0$.[3]



FIGURE 15.2. Tetrahedron natural coordinates: ζ_1, ζ_2, ζ_3, ζ_4.

§15.2.2. Tetrahedral Coordinates

The set of tetrahedral coordinates ζ_1, ζ_2, ζ_3, ζ_4 is the three-dimensional analog of the triangular coordinate set discussed in Chapter 15 of IFEM. The value of ζ_i is one at corner i, zero at the other 3 corners (*i.e.* on the opposite face) and varies linearly as one traverses the distance from the corner to the face. The sum of the four coordinates is identically one:

$$\zeta_1 + \zeta_2 + \zeta_3 + \zeta_4 = 1. \tag{15.3}$$

Any function *linear* in x, y, z, say $F(x, y, z)$, that takes the values F_i ($i = 1, 2, 3, 4$) at the corners may be interpolated in terms of the tetrahedron coordinates as

$$F(\zeta_1,\ \zeta_2,\ \zeta_3,\ \zeta_4) = F_1\zeta_1 + F_2\zeta_2 + F_3\zeta_3 + F_4\zeta_4 = F_i\zeta_i. \tag{15.4}$$

Example 15.1. Suppose that $F(x, y, z) = 4x + 9y - 8z + 3$ and that the coordinates of corners 1,2,3,4 are $(0, 0, 0)$, $(1, 0, 0)$, $(0, 1, 0)$ and $(0, 0, 1)$, respectively. The values of F at the corners are $F_1 = 3$, $F_2 = 7$, $F_3 = 12$ and $F_4 = -5$. Consequently $F(\zeta_1, \zeta_2, \zeta_3, \zeta_4) = 3\zeta_1 + 7\zeta_2 + 12\zeta_3 - 5\zeta_4$.

[3] The tetrahedron volume can be zero only if the four corners are coplanar. This case will be excluded.

§15.2.3. Coordinate Transformations

The geometric definition of the element in terms of these coordinates is obtained by applying the geometry definition (15.4) to x, y and z, and appending the sum-of-coordinates constraint (15.3):

cartesian coord

$$\begin{bmatrix} 1 \\ x \\ y \\ z \end{bmatrix} = \begin{bmatrix} 1 & 1 & 1 & 1 \\ x_1 & x_2 & x_3 & x_4 \\ y_1 & y_2 & y_3 & y_4 \\ z_1 & z_2 & z_3 & z_4 \end{bmatrix} \begin{bmatrix} \zeta_1 \\ \zeta_2 \\ \zeta_3 \\ \zeta_4 \end{bmatrix}. \tag{15.5}$$

tet coord

Inverting this relation gives

a_i, b_i,c_i are function of cartesian coordinate of vertices

$$\begin{bmatrix} \zeta_1 \\ \zeta_2 \\ \zeta_3 \\ \zeta_4 \end{bmatrix} = \frac{1}{6\mathcal{V}} \begin{bmatrix} 6\mathcal{V}_1 & a_1 & b_1 & c_1 \\ 6\mathcal{V}_2 & a_2 & b_2 & c_2 \\ 6\mathcal{V}_3 & a_3 & b_3 & c_3 \\ 6\mathcal{V}_4 & a_4 & b_4 & c_4 \end{bmatrix} \begin{bmatrix} 1 \\ x \\ y \\ z \end{bmatrix}, \tag{15.6}$$

where the coefficients of this matrix can be calculated by forming the adjoints of the matrix in (15.5).

Remark 15.1. The values of a_i, b_i and c_i obtained by explicit inversion are

$$\begin{aligned} a_1 &= y_2 z_{43} - y_3 z_{42} + y_4 z_{32}, & b_1 &= -x_2 z_{43} + x_3 z_{42} - x_4 z_{32}, & c_1 &= x_2 y_{43} - x_3 y_{42} + x_4 y_{32}, \\ a_2 &= -y_1 z_{43} + y_3 z_{41} - y_4 z_{31}, & b_2 &= x_1 z_{43} - x_3 z_{41} + x_4 z_{31}, & c_2 &= -x_1 y_{43} + x_3 y_{41} - x_4 y_{31}, \\ a_3 &= y_1 z_{42} - y_2 z_{41} + y_4 z_{21}, & b_3 &= -x_1 z_{42} + x_2 z_{41} - x_4 z_{21}, & c_3 &= x_1 y_{42} - x_2 y_{41} + x_4 y_{21}, \\ a_4 &= -y_1 z_{32} + y_2 z_{31} - y_3 z_{21}. & b_4 &= x_1 z_{32} - x_2 z_{31} + x_3 z_{21}, & c_4 &= -x_1 y_{32} + x_2 y_{31} - x_3 y_{21}. \end{aligned} \tag{15.7}$$

in which the abbreviations $x_{ij} = x_i - x_j$, $y_{ij} = y_i - y_j$ and $z_{ij} = z_i - z_j$ are used. The volume is given explicitly by

$$6\mathcal{V} = x_{21}(y_{31} z_{41} - y_{41} z_{31}) + y_{21}(x_{41} z_{31} - x_{31} z_{41}) + z_{21}(x_{31} y_{41} - x_{41} y_{31}). \tag{15.8}$$

The values of $\mathcal{V}_i$ are of no interest in what follows.

§15.2.4. *Geometric Interpretation

Figure 15.2 illustrates two geometric interpretation of coordinate ζ_1. In Figure 15.2(a), $\zeta_1 = C$, where C is a number between 0 and 1, is the equation of a plane parallel to the face 234. The plane coincides with that face if $\zeta_1 = 0$, it passes through corner node 1 if $\zeta_1 = 1$, and is interpolated linearly in between.

Figure 15.2(b) illustrates another interpretation that appears in many FEM books. Consider a point P of coordinates $(\zeta_1,\ \zeta_2,\ \zeta_3,\ \zeta_4)$ inside the tetrahedron. Joining P to the corners we obtain four sub-tetrahedra $234P$, $341P$, $412P$ and $123P$, whose volumes are $\mathcal{V}_1$, $\mathcal{V}_2$, $\mathcal{V}_3$ and $\mathcal{V}_4$, respectively. Then ζ_i is the ratio $\mathcal{V}_i/\mathcal{V}$. Figure 15.2(b) pictures the sub-tetrahetron 234P of volume $\mathcal{V}_1$ On account of this relation, tetrahedral coordinates are also called volume coordinates.

think of area(tet) = 1/3 Ah, V_1 and V differs in h only

Remark 15.2. The interpretation as volume coordinates only holds for the tetrahedron defined by 4 corner nodes. It fails for higher order tetrahedra defined by additional nodes (e.g., midpoints). For this reason, the second intepretation, as well as the name "volume coordinates," will not be used here.

§15.2.5. Partial Derivatives

From equations (15.5) and (15.6) we can easily find the following relations for the partial derivatives of Cartesian and tetrahedral coordinates

read off from the (tet <-> cartesian) coordinate transformation matrix

$$\frac{\partial x}{\partial \zeta_i} = x_i, \quad \frac{\partial y}{\partial \zeta_i} = y_i, \quad \frac{\partial z}{\partial \zeta_i} = z_i. \tag{15.9}$$

$$6\mathcal{V}\frac{\partial \zeta_i}{\partial x} = a_i, \quad 6\mathcal{V}\frac{\partial \zeta_i}{\partial y} = b_i, \quad 6\mathcal{V}\frac{\partial \zeta_i}{\partial z} = c_i. \tag{15.10}$$

The derivatives of a function $F(\zeta_1, \zeta_2, \zeta_3, \zeta_4)$ with respect to the Cartesian coordinates follows from (15.10) and the chain rule:

$$\begin{aligned}
\frac{\partial F}{\partial x} &= \frac{\partial F}{\partial \zeta_i}\frac{\partial \zeta_i}{\partial x} = \frac{1}{6\mathcal{V}}\left(\frac{\partial F}{\partial \zeta_1}a_1 + \frac{\partial F}{\partial \zeta_2}a_2 + \frac{\partial F}{\partial \zeta_3}a_3 + \frac{\partial F}{\partial \zeta_4}a_4\right) = \frac{1}{6\mathcal{V}}\frac{\partial F}{\partial \zeta_i}a_i. \\
\frac{\partial F}{\partial y} &= \frac{\partial F}{\partial \zeta_i}\frac{\partial \zeta_i}{\partial y} = \frac{1}{6\mathcal{V}}\left(\frac{\partial F}{\partial \zeta_1}b_1 + \frac{\partial F}{\partial \zeta_2}b_2 + \frac{\partial F}{\partial \zeta_3}b_3 + \frac{\partial F}{\partial \zeta_4}b_4\right) = \frac{1}{6\mathcal{V}}\frac{\partial F}{\partial \zeta_i}b_i. \\
\frac{\partial F}{\partial z} &= \frac{\partial F}{\partial \zeta_i}\frac{\partial \zeta_i}{\partial z} = \frac{1}{6\mathcal{V}}\left(\frac{\partial F}{\partial \zeta_1}c_1 + \frac{\partial F}{\partial \zeta_2}c_2 + \frac{\partial F}{\partial \zeta_3}c_3 + \frac{\partial F}{\partial \zeta_4}c_4\right) = \frac{1}{6\mathcal{V}}\frac{\partial F}{\partial \zeta_i}c_i.
\end{aligned} \tag{15.11}$$

§15.3. The Linear Tetrahedron

The simplest tetrahedron finite element for problems of variational order $m = 1$ is the four-node tetrahedron with *linear shape functions.* The shape functions are simply the tetrahedral coordinates: $N_i = \zeta_i$, $i = 1, 2, 3, 4$. This finite element is derived now for the elasticity problem, using the Total Potential Energy principle as source variational form.

§15.3.1. Displacement Interpolation

The displacement field over the tetrahedron is defined by the three components u_x, u_y and u_z. These are linearly interpolated over the element from their nodal values

$$\begin{bmatrix} u_x \\ u_y \\ u_z \end{bmatrix} = \begin{bmatrix} u_{x1} & u_{x2} & u_{x3} & u_{x4} \\ u_{y1} & u_{y2} & u_{y3} & u_{y4} \\ u_{z1} & u_{z2} & u_{z3} & u_{z4} \end{bmatrix} \begin{bmatrix} \zeta_1 \\ \zeta_2 \\ \zeta_3 \\ \zeta_4 \end{bmatrix}. \tag{15.12}$$

Putting this together with the geometric definition (15.4) we have the isoparametric definition of the 4-node tetrahedron as an elasticity element:

$$\begin{bmatrix} 1 \\ x \\ y \\ z \\ u_x \\ u_y \\ u_z \end{bmatrix} = \begin{bmatrix} 1 & 1 & 1 & 1 \\ x_1 & x_2 & x_3 & x_4 \\ y_1 & y_2 & y_3 & y_4 \\ z_1 & z_2 & z_3 & z_4 \\ u_{x1} & u_{x2} & u_{x3} & u_{x4} \\ u_{y1} & u_{y2} & u_{y3} & u_{y4} \\ u_{z1} & u_{z2} & u_{z3} & u_{z4} \end{bmatrix} \begin{bmatrix} \zeta_1 \\ \zeta_2 \\ \zeta_3 \\ \zeta_4 \end{bmatrix}. \tag{15.13}$$

§15.3.2. The Strain Field

The strain field within the element is strongly connected to the displacement by the strain-displacement equations, which in indicial notation read

$$e_{ij} = \tfrac{1}{2}(u_{i,j} + u_{j,i}). \tag{15.14}$$

We transliterate this to matrix notation as follows. First, the six independent components of the stress tensor are arranged into a 6-component strain vector as follows:

$$\begin{aligned} e &= [\, e_{11} \quad e_{22} \quad e_{33} \quad 2e_{12} \quad 2e_{23} \quad 2e_{31} \,]^T \\ &= [\, e_{xx} \quad e_{yy} \quad e_{zz} \quad \gamma_{xy} \quad \gamma_{yz} \quad \gamma_{zy} \,]^T \end{aligned} . \tag{15.15}$$

The second expression shows the engineering notation for the shear strains. Second, displacement components u_1, u_2 and u_3 are rewritten as u_x, u_y and u_z, collected into a vector and linked to the displacement field by (15.14):

$$\mathbf{e} = \begin{bmatrix} e_{xx} \\ e_{yy} \\ e_{zz} \\ 2e_{xy} \\ 2e_{yz} \\ 2e_{zx} \end{bmatrix} = \begin{bmatrix} \partial/\partial x & 0 & 0 \\ 0 & \partial/\partial y & 0 \\ 0 & 0 & \partial/\partial z \\ \partial/\partial y & \partial/\partial x & 0 \\ 0 & \partial/\partial z & \partial/\partial y \\ \partial/\partial z & 0 & \partial/\partial x \end{bmatrix} \begin{bmatrix} u_x \\ u_y \\ u_z \end{bmatrix} = \mathbf{D}\,\vec{\mathbf{u}}. \tag{15.16}$$

Combining this with (15.12) and using the differentiation rules (15.11) we obtain the matrix relation between strains and nodal displacements:

$$\mathbf{e} = \mathbf{B}\,\mathbf{u}^e. \tag{15.17}$$

If the element nodal displacement vector is arranged component-wise:

$$\mathbf{u}^e = [\, u_{x1} \quad u_{x2} \quad u_{x3} \quad u_{x4} \quad u_{y1} \quad u_{y2} \quad \cdots \quad u_{z4} \,]^T , \tag{15.18}$$

the matrix $\mathbf{B}$ has the following configuration

$$\mathbf{B} = \frac{1}{6\mathcal{V}} \begin{bmatrix} a_1 & a_2 & a_3 & a_4 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & b_1 & b_2 & b_3 & b_4 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & c_1 & c_2 & c_3 & c_4 \\ b_1 & b_2 & b_3 & b_4 & a_1 & a_2 & a_3 & a_4 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & c_1 & c_2 & c_3 & c_4 & b_1 & b_2 & b_3 & b_4 \\ c_1 & c_2 & c_3 & c_4 & 0 & 0 & 0 & 0 & a_1 & a_2 & a_3 & a_4 \end{bmatrix}. \tag{15.19}$$

If the node displacements are arranged node-wise: node-wise displacement field

$$\mathbf{u}^e = [\, u_{x1} \quad u_{y1} \quad u_{z1} \quad u_{x2} \quad u_{y2} \quad u_{z2} \quad \cdots \quad u_{z4} \,]^T , \tag{15.20}$$

the columns of $\mathbf{B}$ must be re-shuffled to yield

$$\mathbf{B} = \frac{1}{6\mathcal{V}} \begin{bmatrix} a_1 & 0 & 0 & a_2 & 0 & 0 & a_3 & 0 & 0 & a_4 & 0 & 0 \\ 0 & b_1 & 0 & 0 & b_2 & 0 & 0 & b_3 & 0 & 0 & b_4 & 0 \\ 0 & 0 & c_1 & 0 & 0 & c_2 & 0 & 0 & c_3 & 0 & 0 & c_4 \\ b_1 & a_1 & 0 & b_2 & a_2 & 0 & b_3 & a_3 & 0 & b_4 & a_4 & 0 \\ 0 & c_1 & b_1 & 0 & c_2 & b_2 & 0 & c_3 & b_3 & 0 & c_4 & b_4 \\ c_1 & 0 & a_1 & c_2 & 0 & a_2 & c_3 & 0 & a_3 & c_4 & 0 & a_4 \end{bmatrix}. \tag{15.21}$$

The node-wise arrangement (15.20) of $\mathbf{u}^e$ is more common in practice because it facilitates the assembly process.

Note that both matrices (15.19) and (15.21) are constant over the element.

§15.3.3. The Stress Field

The stress field is related to the stress field by the strong connection

$$\sigma_{ij} = E_{ijk\ell} e_{k\ell} \tag{15.22}$$

To convert this to matrix notation we rearrange the 6 independent stress components to correspond to the strains (15.12) and link them by a 6 × 6 matrix of elastic moduli:

$$\begin{aligned} \boldsymbol{\sigma} &= [\,\sigma_{11} \quad \sigma_{22} \quad \sigma_{33} \quad \sigma_{12} \quad \sigma_{23} \quad \sigma_{31}\,]^T = \\ &= [\,\sigma_{xx} \quad \sigma_{yy} \quad \sigma_{zz} \quad \sigma_{xy} \quad \sigma_{yz} \quad \sigma_{zx}\,]^T \end{aligned} \tag{15.23}$$

If the material is linearly elastic and no initial strains are considered, the constitutive equation may be compactly expressed as

$$\boldsymbol{\sigma} = \mathbf{E}\,\mathbf{e}. \tag{15.24}$$

where the elasticity matrix $\mathbf{E}$ is symmetric. For a general anisotropic material the expanded form of (15.24) is

$$\begin{bmatrix} \sigma_{xx} \\ \sigma_{yy} \\ \sigma_{zz} \\ \sigma_{xy} \\ \sigma_{yz} \\ \sigma_{zx} \end{bmatrix} = \begin{bmatrix} E_{11} & E_{12} & E_{13} & E_{14} & E_{15} & E_{16} \\ & E_{22} & E_{23} & E_{24} & E_{25} & E_{26} \\ & & E_{33} & E_{34} & E_{35} & E_{36} \\ & & & E_{44} & E_{45} & E_{46} \\ & & & & E_{55} & E_{56} \\ & \text{symm} & & & & E_{66} \end{bmatrix} \begin{bmatrix} e_{xx} \\ e_{yy} \\ e_{zz} \\ 2e_{xy} \\ 2e_{yz} \\ 2e_{zx} \end{bmatrix}, \tag{15.25}$$

in which E_{ij} are constitutive moduli. For an isotropic material of elastic modulus E and Poisson's ratio ν the foregoing relation simplifies to

$$\begin{bmatrix} \sigma_{xx} \\ \sigma_{yy} \\ \sigma_{zz} \\ \sigma_{yz} \\ \sigma_{zx} \\ \sigma_{xy} \end{bmatrix} = \frac{E}{(1+\nu)(1-2\nu)} \begin{bmatrix} 1-\nu & \nu & \nu & 0 & 0 & 0 \\ \nu & 1-\nu & \nu & 0 & 0 & 0 \\ \nu & \nu & 1-\nu & 0 & 0 & 0 \\ 0 & 0 & 0 & \frac{1}{2}-\nu & 0 & 0 \\ 0 & 0 & 0 & 0 & \frac{1}{2}-\nu & 0 \\ 0 & 0 & 0 & 0 & 0 & \frac{1}{2}-\nu \end{bmatrix}. \tag{15.26}$$

§15.4. The Element Stiffness Matrix

Introducing $\mathbf{e} = \mathbf{B}\mathbf{u}$ and $\boldsymbol{\sigma} = \mathbf{E}\mathbf{e}$ into the strain energy functional restricted to the element volume and rendering the resulting algebraic form stationary with respect to the node displacements $\mathbf{u}^e$ we get the usual expression for the element stiffness matrix

$$\boxed{\mathbf{K}^e = \int_{V^e} \mathbf{B}^T \, \mathbf{E} \, \mathbf{B} \, dV.} \tag{15.27}$$

Assuming that the elastic moduli are constant inside the element, the foregoing integrand is constant because matrix $\mathbf{B}$ is constant — cf. (15.19) or (15.21). Consequently

$$\boxed{\mathbf{K}^e = \mathcal{V} \, \mathbf{B}^T \, \mathbf{E} \, \mathbf{B}.} \qquad \text{ehh} \tag{15.28}$$

This stiffness matrix is 12×12. It can be directly evaluated in closed form using the above expression or, equivalently, by a one-point (centroid) integration rule.

§15.5. The Consistent Node Force Vector

A terahedral mesh may be subjected to given body forces in the volume and/or specified boundary tractions. Both have to be converted to node forces through an energy-based lumping procedure.

§15.5.1. Body Forces

Consider a body force field over the element, such as gravity of centrifugal forces, defined by its components

$$\mathbf{b} = [\, b_x \quad b_y \quad b_z \,]^T \,. \tag{15.29}$$

Inserting this into the TPE principle, the body force contribution gives

$$\boxed{\mathbf{f}^e = \int_{V^e} \mathbf{N}^T \, \mathbf{b} \; dV.} \tag{15.30}$$

Here $\mathbf{N}$ is the 3×12 matrix of shape functions that relates element field displacements to node displacements:

$$\vec{\mathbf{u}} = \begin{bmatrix} u_x \\ u_y \\ u_z \end{bmatrix} = \mathbf{N} \, \mathbf{u}^e. \qquad \text{N: node-wise -> cartesian} \tag{15.31}$$

For the component-wise node displacement ordering (15.18),

$$\mathbf{N} = \begin{bmatrix} \zeta_1 & \zeta_2 & \zeta_3 & \zeta_4 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & \zeta_1 & \zeta_2 & \zeta_3 & \zeta_4 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & \zeta_1 & \zeta_2 & \zeta_3 & \zeta_4 \end{bmatrix} \tag{15.32}$$

For the node-wise displacement ordering (15.20),

$$\mathbf{N} = \begin{bmatrix} \zeta_1 & 0 & 0 & \zeta_2 & 0 & 0 & \zeta_3 & 0 & 0 & \zeta_4 & 0 & 0 \\ 0 & \zeta_1 & 0 & 0 & \zeta_2 & 0 & 0 & \zeta_3 & 0 & 0 & \zeta_4 & 0 \\ 0 & 0 & \zeta_1 & 0 & 0 & \zeta_2 & 0 & 0 & \zeta_3 & 0 & 0 & \zeta_4 \end{bmatrix} \tag{15.33}$$

Even if the body forces are constant the integral is not constant over the element. Some useful formulae for such calculations are

$$\int_{V^e} \zeta_i \, dV = \tfrac{1}{4}\mathcal{V}, \tag{15.34}$$

and

$$\int_{V^e} \zeta_i \zeta_j \, dV = \begin{cases} \frac{1}{10}\mathcal{V} & \text{if } i = j, \\ \frac{1}{20}\mathcal{V} & \text{if } i \neq j. \end{cases} \tag{15.35}$$

The general rule for such integrals, which can be derived from the Beta function, is

$$\int_{V^e} \zeta_1^i \, \zeta_2^j \, \zeta_3^k \, \zeta_4^\ell \, dV = \frac{i!\; j!\; k!\; \ell!}{(i+j+k+\ell+3)!} 6\mathcal{V}. \tag{15.36}$$

in which i, j, k and ℓ are nonnegative integers. This formula is only valid for tetrahedra with planar faces.

§15.5.2. Surface Tractions

The most practically important case is that of surface tractions normal to an element face. This models the effect of pressure loads. The calculation of node forces for the case of a constant pressure acting on a tetrahedron face is the matter of one exercise.

§15.5.3. Element Implementation

The implementation of the linear tetrahedron in any programming language is very simple. An implementation in the form of a *Mathematica* module is shown in Figure 15.3. The module is invoked as

```
Ke=Trig3IsoPMembraneStiffness[ncoor,Emat,{ },options];
```
(15.37)

The arguments are

`encoor` Element node coordinates, arranged as a list: `{{x1,y1,z1},{x2,y2,z2},{x3,y3,z3},{x4,y4,z4}}`.

`Emat` A two-dimensional list storing the 6 × 6 matrix of elastic moduli as `{{E11,E12,E13,E14,E15,E16}, ... {E61,E62,E63,E64,E65,E66}}`.

`options` A list of formation options. For this element it is simply `{ numer }`, where `numer` is a logical flag. Flag is `True` to request floating point numeric work, `False` to request exact calculations.

The third argument is a placeholder and should be set to the empty list { }.

The stiffness module calls module `IsoTetr4ShapeFunCarDer`, which is listed in Figure 15.4, to get the shape function Cartesian partial derivatives. These return in the 12 x 1 arrays `Bx`, `By` and `Bz`. The module also returns the jacobian determinant `Jdet`, which is six times the element volume.

Module `IsoTetr4ShapeFunCarDer` is written in a more complicated style than needed for this particular element. For example `J11` is simply `x1`, etc. It is actually configured to serve as a shape function derivative "template" for more refined tetrahedron elements, as described in the next Chapter. r

```
IsoTetr4Stiffness[ncoor_,Emat_,{},options_]:= Module[{i,n=4,nf=12,
   k,c,w,Jdet,zetalist,xyzlist,numer,Bx,By,Bz,Be,Ke},
  If [Length[options]>0, numer=options[[1]]];
  xyzlist={Table[ncoor[[i,1]],{i,n}],Table[ncoor[[i,2]],{i,n}],
           Table[ncoor[[i,3]],{i,n}]}; Ke=Table[0,{nf},{nf}];
  {Bx,By,Bz,Jdet}=IsoTetr4ShapeFunCarDer[xyzlist,{},numer];
  Be={Flatten[Table[{Bx[[i]],0,      0      },{i,n}]],
      Flatten[Table[{0,      By[[i]],0      },{i,n}]],
      Flatten[Table[{0,      0,      Bz[[i]]},{i,n}]],
      Flatten[Table[{By[[i]],Bx[[i]],0      },{i,n}]],
      Flatten[Table[{0,      Bz[[i]],By[[i]]},{i,n}]],
      Flatten[Table[{Bz[[i]],0,      Bx[[i]]},{i,n}]]};
   Ke=(Jdet/6)*Transpose[Be].(Emat.Be);
   If [!numer,Ke=Simplify[Ke]]; Return[Ke]
   ];
```

FIGURE 15.3. Module to form the stiffness matrix of a linear tetrahedron (Tetr4) and outputs.

```
IsoTetr4ShapeFunCarDer[{xn_,yn_,zn_},zetalist_,numer_]:=
  Module[{dNz1,dNz2,dNz3,dNz4,Jmat,J11,J12,J13,J14,
   J21,J22,J23,J24,J31,J32,J33,J34,Jinv,Jdet,Bx,By,Bz},
   {dNz1,dNz2,dNz3,dNz4}={{1,0,0,0},{0,1,0,0},{0,0,1,0},{0,0,0,1}};
   J11=dNz1.xn; J12=dNz2.xn; J13=dNz3.xn;  J14=dNz4.xn;
   J21=dNz1.yn; J22=dNz2.yn; J23=dNz3.yn;  J24=dNz4.yn;
   J31=dNz1.zn; J32=dNz2.zn; J33=dNz3.zn;  J34=dNz4.zn;
   Jmat={{1,1,1,1},{J11,J12,J13,J14},
         {J21,J22,J23,J24},{J31,J32,J33,J34}};
   Jdet=(J13*J22-J12*J23+J14*J23-J14*J22+J12*J24-J13*J24)*J31-
        (J13*J21-J11*J23+J14*J23-J14*J21+J11*J24-J13*J24)*J32+
        (J12*J21-J11*J22+J14*J22-J14*J21+J11*J24-J12*J24)*J33-
        (J12*J21-J11*J22+J13*J22-J13*J21+J11*J23-J12*J23)*J34;
   Jinv={{J22*(J34-J33)-J23*(J34-J32)+J24*(J33-J32),
         -J12*(J34-J33)+J13*(J34-J32)-J14*(J33-J32),
          J12*(J24-J23)-J13*(J24-J22)+J14*(J23-J22)},
        {-J21*(J34-J33)+J23*(J34-J31)-J24*(J33-J31),
          J11*(J34-J33)-J13*(J34-J31)+J14*(J33-J31),
         -J11*(J24-J23)+J13*(J24-J21)-J14*(J23-J21)},
        { J21*(J34-J32)-J22*(J34-J31)+J24*(J32-J31),
         -J11*(J34-J32)+J12*(J34-J31)-J14*(J32-J31),
          J11*(J24-J22)-J12*(J24-J21)+J14*(J22-J21)},
        {-J21*(J33-J32)+J22*(J33-J31)-J23*(J32-J31),
          J11*(J33-J32)-J12*(J33-J31)+J13*(J32-J31),
         -J11*(J23-J22)+J12*(J23-J21)-J13*(J22-J21)}};
   {Bx,By,Bz}=Transpose[Jinv].{dNz1,dNz2,dNz3,dNz4}/Jdet;
   Return[{Bx,By,Bz,Jdet}]
   ];
```

FIGURE 15.4. Module to compute shape function partial derivatives for linear tetrahedron (Tetr4). As noted in the text, it is deliberated written in a more general fashion than needed for this particular element.

```
ClearAll[Em,ν]; Em=96; ν=1/3;
Emat=Em/((1+ν)*(1-2*ν))*{{1-ν,ν,ν,0,0,0},
  {ν,1-ν,ν,0,0,0},{ν,ν,1-ν,0,0,0},{0,0,0,1/2-ν,0,0},
  {0,0,0,0,1/2-ν,0},{0,0,0,0,0, 1/2-ν}};
Print["Emat=",Emat//MatrixForm];
ncoor={{2,3,4},{6,3,2},{2,5,1},{4,3,6}};
Ke=IsoTetr4Stiffness[ncoor,Emat,{},{False}];
Print["Ke=",Ke//MatrixForm];
Print["eigs of Ke=",Chop[Eigenvalues[N[Ke]]]];
```

FIGURE 15.5. Test statements to exercise the module of Figure 15.3.

$$\text{Emat} = \begin{pmatrix} 144 & 72 & 72 & 0 & 0 & 0 \\ 72 & 144 & 72 & 0 & 0 & 0 \\ 72 & 72 & 144 & 0 & 0 & 0 \\ 0 & 0 & 0 & 36 & 0 & 0 \\ 0 & 0 & 0 & 0 & 36 & 0 \\ 0 & 0 & 0 & 0 & 0 & 36 \end{pmatrix}$$

$$\text{Ke} = \begin{pmatrix} 149 & 108 & 24 & -1 & 6 & 12 & -54 & -48 & 0 & -94 & -66 & -36 \\ 108 & 344 & 54 & -24 & 104 & 42 & -24 & -216 & -12 & -60 & -232 & -84 \\ 24 & 54 & 113 & 0 & 30 & 35 & 0 & -24 & -54 & -24 & -60 & -94 \\ -1 & -24 & 0 & 29 & -18 & -12 & -18 & 24 & 0 & -10 & 18 & 12 \\ 6 & 104 & 30 & -18 & 44 & 18 & 12 & -72 & -12 & 0 & -76 & -36 \\ 12 & 42 & 35 & -12 & 18 & 29 & 0 & -24 & -18 & 0 & -36 & -46 \\ -54 & -24 & 0 & -18 & 12 & 0 & 36 & 0 & 0 & 36 & 12 & 0 \\ -48 & -216 & -24 & 24 & -72 & -24 & 0 & 144 & 0 & 24 & 144 & 48 \\ 0 & -12 & -54 & 0 & -12 & -18 & 0 & 0 & 36 & 0 & 24 & 36 \\ -94 & -60 & -24 & -10 & 0 & 0 & 36 & 24 & 0 & 68 & 36 & 24 \\ -66 & -232 & -60 & 18 & -76 & -36 & 12 & 144 & 24 & 36 & 164 & 72 \\ -36 & -84 & -94 & 12 & -36 & -46 & 0 & 48 & 36 & 24 & 72 & 104 \end{pmatrix}$$

eigs of Ke = {777.175, 201.363, 197.273, 42.9431, 21.3643, 19.8821, 0, 0, 0, 0, 0, 0}

FIGURE 15.6. Results from running test of Figure 15.5.

The stiffness module is exercised by the statements listed in Figure 15.5, which forms a tetrahedron with corner coordinates $\{x_1, y_1, z_1\} = \{2, 3, 4\}$, $\{x_2, y_2, z_2\} = \{6, 3, 2\}$, $\{x_3, y_3, z_3\} = \{2, 5, 1\}$ and $\{x_4, y_4, z_4\} = \{4, 3, 6\}$. Its volume is $+24$. The material is isotropic with elastic modulus $E = 96$ and Poisson's ratio $\nu = 1/3$. The results are shown in Figure 15.6. The computation of stiffness matrix eigenvalues is always a good programming test, since 6 eigenvalues (associated with rigid body modes) must be exactly zero and the other 6 real and positive. This is verified by the results.

Homework Exercises for Chapter 15

The Linear Tetrahedron

EXERCISE 15.1 [A:5] The tetrahedron element does not have fabrication properties, such as the thickness in the case of a plane stress element. Why?

EXERCISE 15.2 [A:15] Work out the formulas for a_i, b_i, c_i in terms of the corner coordinates x_i, y_i and z_i ($i = 1, 2, 3, 4$). Then write a compact formula for the volume $\mathcal{V}$. Hint: use the following script:

```
J={{1,1,1,1},{x1,x2,x3,x4},{y1,y2,y3,y4},{z1,z2,z3,z4}};
V6=Det[J]; Jinv=Simplify[Inverse[J]*V6];
{{R1,a1,b1,c1},{R2,a2,b2,c2},{R3,a3,b3,c3},{R4,a4,b4,c4}}=Jinv;
Print["{a1,a2,a3,a4}*V6=",{a1,a2,a3,a4}];
Print["{b1,b2,b3,b4}*V6=",{b1,b2,b3,b4}];
Print["{c1,c2,c3,c4}*V6=",{c1,c2,c3,c4}];
Print["V6=",V6];
```

EXERCISE 15.3 [A:20] Work out by hand the consistent node force vector $\mathbf{f}^e$ for the body force system $b_x = b_{x1}\zeta_1 + b_{x2}\zeta_2 + b_{x3}\zeta_3 + b_{x4}\zeta_4$, $b_y = 0$, $b_z = 0$, in which b_{xi} are given values at the nodes. Hint: use the integration rule (15.36). Specialize the result to $b_{x1} = b_{x2} = b_{x3} = b_{x4} = b_x$.

EXERCISE 15.4 [A:25] Face 1-2-3 of the 4-node tetrahedron is under pressure p acting normal to the face (positive pressure: $+p$ means it points into the body). Compute $\mathbf{f}^e$. [*Hint:* find the direction cosines n_j (needed to get the prescribed surface tractions $\hat{t}_i$) of the unit normal by developing the normal equation of plane 1-2-3].